
 

SULAMA SISTEMI 

FULL PROGRAM 
 

 

  SULAMA SİSTEMİ V2.5 - KULLANICI KILAVUZU 

Selçuk DİKİCİ Akıllı Tarım Teknolojileri Web: selcukdikici.com.tr 

 

1. BAŞLANGIÇ VE EKRAN TANITIMI 

Cihaza güç verdiğinizde karşınıza bir yükleme barı ve karşılama ekranı gelir. Sistem açıldığında ana 

ekranda iki satırlı bilgi akışı başlar: 

• Üst Satır: Güncel saat, sıcaklık (T), nem (H) ve rölelerin (R1-R2) anlık durumu. 

• Alt Satır: Cihazda kurulu olan aktif programlar (DELAY1.1, ISI-AKTIF vb.) ve hızlı komut 

rehberi. 

 

2. HIZLI KOMUTLAR VE KISAYOLLAR 

Cihazı yönetmek için tuş takımını aşağıdaki gibi kullanabilirsiniz: 

• [*] Tuşu: 1. Röleyi (Vana/Pompa) anlık açar veya kapatır. 

• [#] Tuşu: 2. Röleyi anlık açar veya kapatır. 

• [0] Tuşu: ACİL DURUM MODU. Her iki röleyi de aynı anda aktif eder. 

• [A] Tuşu: Yeni sulama programı ekleme menüsü. 

• [B] Tuşu: Mevcut bir programı silme menüsü. 

• [C] Tuşu: Menülerden çıkış veya hatalı girişi iptal etme. 

 

3. SİSTEM AYARLARI (ŞİFRELİ MODLAR) 

Sistemin beynine hükmetmek için aşağıdaki özel kodları ana ekrandayken tuşlayın: 

      Saat ve Tarih Ayarı: 252512 

Sensörlerin ve zamanlayıcıların doğru çalışması için sistem saati güncel olmalıdır. 

https://selcukdikici.com.tr/


1. Kodu tuşlayın. 

2. Gün: 1 (Pazartesi) ile 7 (Pazar) arası bir rakam girin. 

3. Saat & Dakika: 2'şer hane olarak girin (Örn: 09:05). 

   Akıllı Isı Kontrolü: 242424 

Bitkilerin aşırı sıcakta korunması için kullanılır. 

• Bir eşik değeri girin (Örn: 32). Sıcaklık bu değerin üzerine çıkarsa ve saat 22:00-23:59 

arasındaysa sistem otomatik sulama başlatır. 

   Akıllı Nem Kontrolü: 232323 

Toprağın veya ortamın kurumasını engeller. 

• Bir nem eşiği girin (Örn: 40). Nem bu değerin altına düşerse ve saat 22:00-23:59 arasındaysa 

sistem otomatik sulama başlatır. 

 

4. PROGRAMLAMA REHBERİ (A TUŞU) 

Sistemde toplam 10 adet bağımsız sulama hücresi (R1 için 5, R2 için 5) bulunur. 

1. A tuşuna basın. 

2. Röle: 1 veya 2'yi seçin. 

3. Hücre: 1 ile 5 arası bir numara seçin. 

4. Açılış Zamanı: Önce günü (1-7), sonra saati ve dakikayı girin. 

5. Kapanış Zamanı: Sulamanın biteceği günü, saati ve dakikayı girin. 

o Not: Açılış ve kapanış günleri farklı olabilir (Örn: Pazartesi aç - Salı kapat). 

 

5. GÜVENLİK VE SIFIRLAMA 

• Hata Uyarıları: Eğer zaman veya gün sınırları dışında bir rakam girerseniz ekran "LIMIT 

DISI!" uyarısı vererek sizi uyarır. 

• Fabrika Ayarları: Tüm hafızayı temizlemek ve şifreleri sıfırlamak için 252525 tuşlayın. Sistem 

her şeyi siler ve kendini yeniden başlatır. 

 

      ÖNEMLİ HATIRLATMALAR 

• Sensör modları (Isı ve Nem) bitkilerin sağlığı için sadece gece saatlerinde (22:00 - 23:59) 

çalışacak şekilde kısıtlanmıştır. 

• Elektrik kesintilerinde ayarlarınız silinmez; sistem EEPROM hafızası sayesinde kaldığı yerden 

devam eder. 

• Ekranda [T] veya [H] harflerini görüyorsanız, akıllı sensör modları görev başındadır. 



 

Sulama Sistemi V2.5 Montaj Kılavuzu 

Bu sistem, hassas sensör okumaları ve 10 farklı zamanlama hücresiyle çalıştığı için bağlantıların 

sağlamlığı çok önemlidir. 

1. Kullanılacak Malzemeler 

• Kontrolcü: Arduino Uno R3 

• Ekran: 1602A LCD + I2C Modülü (4 Pinli olan model) 

• Giriş: 4x4 Membran Tuş Takımı (Keypad) 

• Sensör: DHT11 Isı ve Nem Sensörü 

• Çıkış: 2'li Röle Modülü (5V tetiklemeli) 

• Güç: 9V 2A Adaptör (Sistem ve röleler için) 

 

2. Devre Bağlantı Şeması (Pin Pin Anlatım) 

A. I2C LCD Ekran Bağlantısı 

LCD'nin arkasındaki I2C modülünden çıkan 4 kabloyu Arduino'ya bağla: 

• GND -> Arduino GND 

• VCC -> Arduino 5V 

• SDA -> Arduino A4 

• SCL -> Arduino A5 

B. DHT11 Isı ve Nem Sensörü 

• VCC (+) -> Arduino 5V 

• GND (-) -> Arduino GND 

• DATA (S) -> Arduino A1 (Kodda A1 olarak güncelledik, çakışmayı önler) 

C. 4x4 Keypad (Tuş Takımı) 

Keypad'in 8 adet pini vardır. Soldan sağa doğru sırayla şuraya tak: 

• Pin 1-4 (Satırlar): Arduino D9, D8, D7, D6 

• Pin 5-8 (Sütunlar): Arduino D5, D4, D3, D2 

D. 2'li Röle Modülü 

• VCC -> Arduino 5V 

• GND -> Arduino GND 

• IN1 (Röle 1) -> Arduino D12 



• IN2 (Röle 2) -> Arduino D11 

 

3. Kritik Montaj Notları (Dikkat!) 

• Güç Sorunu: Eğer röleler açıldığında LCD ekranın ışığı gidiyorsa veya karakterler bozuluyorsa, 

Arduino'ya sadece bilgisayarın USB'sinden güç veriyorsun demektir. 9V 2A'lik bir adaptörü 

Arduino'nun güç girişine mutlaka takmalısın. 

• Sensör Sağlığı: DHT11 sensörünü rölelerden ve su pompalarından biraz uzağa monte et ki 

nemden etkilenip yanlış ölçüm yapmasın. 

• Kablolama: Keypad kabloları ince olduğu için gevşeyebilir. Jumper kabloların pinlere tam 

oturduğundan emin ol. 

 

4. Sistemi Kullanma Kılavuzu (Kısayollar) 

Tuş / Kod İşlevi 

A Tuşu Programlama menüsünü açar (10 adet DELAY hücresi). 

B Tuşu Kurulu olan programları siler. 

* Tuşu Röle 1'i manuel açar/kapatır. 

# Tuşu Röle 2'yi manuel açar/kapatır. 

0 Tuşu ACİL DURUM: İki röleyi de anında açar. 

252512 Sistem saatini ve gününü ayarlar. 

242424 Isı Modunu (ISI-AKTIF) eşik değerle kurar. 

232323 Nem Modunu (NEM-AKTIF) eşik değerle kurar. 

252525 Tüm hafızayı (EEPROM) siler ve sistemi resetler. 

 

5. İlk Çalıştırma Testi 

1. Gücü ver, ekranda selcukdikici.com.tr yazısını gör. 

2. Önce 252512 yazıp saati ve günü ayarla (Sensör modları gece 22:00'den sonra çalıştığı için 

saatin doğru olması şart!). 

3. * tuşuna basıp rölelerin "çıt" ettiğini duy. 

4. Alt satırda "A:Prog B:Sil..." yazısının aktığını kontrol et. 

Kurulması Gereken Kütüphane Paketleri 



Arduino IDE içinde Taslak (Sketch) > library ekle > Kütüphaneleri Yönet... yolunu izle ve arama 

çubuğuna şu isimleri yazarak kur: 

1. LiquidCrystal I2C (Ekran Paketi) 

• Arama Terimi: LiquidCrystal I2C 

• Yayımcı: Frank de Brabander (En stabil olan budur). 

• Neden Gerekli? Ekranın I2C modülü üzerinden sadece 2 kablo ile çalışmasını sağlar. 

2. Keypad (Tuş Takımı Paketi) 

• Arama Terimi: Keypad 

• Yayımcı: Mark Stanley, Alexander Brevig. 

• Neden Gerekli? 4x4 tuş takımındaki basılan tuşları algılar ve şifre sistemini yönetir. 

3. DHT Sensor Library (Isı ve Nem Paketi) 

• Arama Terimi: DHT sensor library 

• Yayımcı: Adafruit. 

• Not: Bunu kurarken karşına "Install All" (Hepsini Kur) diye bir seçenek çıkacaktır. Ona 

basmalısın; çünkü bu paket çalışmak için Adafruit Unified Sensor paketine de ihtiyaç duyar. 

 

     Arduino IDE Ayarları (Yükleme Öncesi) 

Kodu yüklemeden önce şu ayarların seçili olduğundan emin ol: 

1. Araçlar > Kart: Arduino Uno 

2. Araçlar > Port: Arduino'nun takılı olduğu port (Örn: COM3 veya COM4). 

3. Araçlar > İşlemci: ATmega328P (Genelde otomatik seçilir). 

 

       Adım Adım Programlama Kılavuzu 

Sistemi yükledikten sonra cihaz üzerinden şu adımları takip ederek kurulumu tamamla: 

Adım 1: Zamanı Ayarla (Kritik!) 

Sensör modlarının gece çalışabilmesi için saatin doğru olması şarttır. 

• Ana ekranda 252512 tuşla. 

• Sırasıyla: Gün (1-7), Saat (00-23) ve Dakika (00-59) bilgilerini gir. 

Adım 2: Sulama Programı Kur (DELAY Modu) 

• A tuşuna bas. 

• Röle Seç: 1 veya 2. 

• Program No: 1 ile 5 arasında bir hücre seç. 



• Zamanlama: Açılış gün/saat/dakika ve Kapanış gün/saat/dakika bilgilerini gir. 

• Örnek: Pazartesi (1) 08:00 aç, Pazartesi (1) 08:30 kapat. 

Adım 3: Akıllı Sensör Modlarını Aktif Et 

• Isı İçin: 242424 tuşla ve bir eşik sıcaklığı gir (Örn: 30). Sıcaklık 30'u geçince ve saat 22:00-

23:59 arasındaysa R1 çalışır. 

• Nem İçin: 232323 tuşla ve nem eşiği gir (Örn: 40). Nem %40'ın altına düşerse ve saat 22:00-

23:59 arasındaysa R1 çalışır. 

Adım 4: Hafızayı Sıfırlama (Fabrika Ayarları) 

• Sistemi tamamen temizlemek istersen 252525 tuşla. Ekran donar, hafıza silinir ve sistem senin 

web sitenle (Selcuk DIKICI) yeniden başlar. 

 

Küçük Bir İpucu: Kodu yüklerken kütüphane hatası alırsan, kütüphane yöneticisinde kurulu paketlerin 

yanında "Update" yazıp yazmadığını kontrol et, varsa güncelle tatlım. 

 

PROGRAM 

#include <Wire.h> 

#include <LiquidCrystal_I2C.h> 

#include <Keypad.h> 

#include <EEPROM.h> 

#include <DHT.h> 

 

// --- Donanım Ayarları --- 

LiquidCrystal_I2C lcd(0x27, 16, 2); 

#define DHTPIN A1 

#define DHTTYPE DHT11 

DHT dht(DHTPIN, DHTTYPE); 

 

const int roleler[2] = {12, 11}; 

const byte ROWS = 4, COLS = 4; 

char keys[ROWS][COLS] = { 

  {'1','2','3','A'}, {'4','5','6','B'}, {'7','8','9','C'}, {'*','0','#','D'} 

}; 



byte rowPins[ROWS] = {9, 8, 7, 6}; 

byte colPins[COLS] = {5, 4, 3, 2}; 

Keypad keypad = Keypad(makeKeymap(keys), rowPins, colPins, ROWS, COLS); 

 

struct Alarm { int acGun, acS, acD, kapGun, kapS, kapD; bool aktif; }; 

Alarm programlar[2][5];  

 

float sicaklikEsik = 0; 

int nemEsik = 0; 

bool sicaklikModu = false, nemModu = false; 

unsigned long eskiMillis = 0, kMillis1 = 0, kMillis2 = 0; 

int saniye = 0, dakika = 0, saat = 12, gun = 1; 

float sonIsi = 0, sonNem = 0; 

String tusKaydi = ""; 

int pos1 = 0, pos2 = 0;  

bool ayarModu = false; 

const char* gunIsimleri[] = {"", "Pzt", "Sal", "Car", "Per", "Cum", "Cmt", "Paz"}; 

 

void(* resetFunc) (void) = 0;  

 

// --- GÜVENLİ VERİ GİRİŞİ --- 

int veriGirisi(const char* mesaj, int hane, int minLimit, int maxLimit) { 

  ayarModu = true; String d = ""; 

  while (true) { 

    lcd.clear(); lcd.print(mesaj); d = ""; 

    while(d.length() < hane) { 

      char k = keypad.getKey(); 

      if(k >= '0' && k <= '9') { d += k; lcd.setCursor(d.length()-1, 1); lcd.print(k); } 

      if(k == 'C') { ayarModu = false; return -1; } 

    } 

    int sonuc = d.toInt(); 



    if (sonuc >= minLimit && sonuc <= maxLimit) { delay(300); ayarModu = false; return sonuc; } 

    else { lcd.clear(); lcd.print(F("LIMIT DISI!")); delay(1500); } 

  } 

} 

 

// --- SİSTEM KONTROLÜ --- 

void sistemKontrol() { 

  if (ayarModu) return; 

  long suanDakika = (long)gun * 1440 + (long)saat * 60 + dakika; 

  bool r1_istek = false, r2_istek = false; 

 

  // Sensör Kontrolü (22:00-23:59) 

  if (saat >= 22 && saat < 24) { 

    if (sicaklikModu && sonIsi >= sicaklikEsik) r1_istek = true; 

    if (nemModu && sonNem <= nemEsik) r1_istek = true; 

  } 

 

  // Program Kontrolü 

  for(int r=0; r<2; r++) { 

    for(int p=0; p<5; p++) { 

      if(programlar[r][p].aktif) { 

        long bas = (long)programlar[r][p].acGun * 1440 + (long)programlar[r][p].acS * 60 + 

programlar[r][p].acD; 

        long bit = (long)programlar[r][p].kapGun * 1440 + (long)programlar[r][p].kapS * 60 + 

programlar[r][p].kapD; 

        bool aktif = (bit < bas) ? (suanDakika >= bas || suanDakika < bit) : (suanDakika >= bas && 

suanDakika < bit); 

        if(aktif) { if(r==0) r1_istek = true; else r2_istek = true; } 

      } 

    } 

  } 

  digitalWrite(roleler[0], r1_istek ? LOW : HIGH); 



  digitalWrite(roleler[1], r2_istek ? LOW : HIGH); 

} 

 

// --- AKTİF PROGRAMLARI VE MODLARI AL --- 

String aktifDurumlariAl() { 

  String text = ""; 

   

  // Önce Sensör Modlarını Ekle 

  if(sicaklikModu) text += "ISI-AKTIF "; 

  if(nemModu) text += "NEM-AKTIF "; 

   

  // Sonra Programlı Adımları Ekle 

  for(int r=0; r<2; r++) { 

    for(int p=0; p<5; p++) { 

      if(programlar[r][p].aktif) { 

        text += "DELAY" + String(r+1) + "." + String(p+1) + "-AKTIF "; 

      } 

    } 

  } 

  return text; 

} 

 

void setup() { 

  Wire.begin(); lcd.init(); lcd.backlight(); dht.begin(); 

  lcd.clear(); lcd.print(F("SELCUK DIKICI")); lcd.setCursor(0,1); lcd.print(F("Tr - V2.5")); delay(2000); 

   

  int adr = 0; 

  for(int r=0; r<2; r++) for(int p=0; p<5; p++) {  

    EEPROM.get(adr, programlar[r][p]);  

    adr += sizeof(Alarm);  

  } 



  for(int i=0; i<2; i++) { pinMode(roleler[i], OUTPUT); digitalWrite(roleler[i], HIGH); } 

  sistemKontrol(); 

} 

 

void loop() { 

  if(millis() - eskiMillis >= 1000) { 

    eskiMillis = millis(); saniye++; 

    if(saniye >= 60) { saniye = 0; dakika++; sistemKontrol(); } 

    if(dakika >= 60) { dakika = 0; saat++; } 

    if(saat >= 24) { saat = 0; if(++gun > 7) gun = 1; } 

  } 

 

  static unsigned long sMillis = 0; 

  if(millis() - sMillis > 2000) { 

    sMillis = millis(); 

    float t = dht.readTemperature(); float h = dht.readHumidity(); 

    if(!isnan(t)) { sonIsi = t; sonNem = h; } 

  } 

 

  if(!ayarModu) { 

    // Üst Satır: Zaman, Isı, Nem ve Röle Durumları 

    if(millis() - kMillis1 > 400) { 

      kMillis1 = millis(); 

      String m1 = String(gunIsimleri[gun]) + " " + 

(saat<10?"0":"")+String(saat)+":"+(dakika<10?"0":"")+String(dakika) +  

                  " T:" + String((int)sonIsi) + "C H:%" + String((int)sonNem) + " | R1:" +  

                  String(digitalRead(roleler[0])==LOW?"ACIK":"KAPALI") + "   "; 

      String g1 = m1.substring(pos1) + m1.substring(0, pos1); 

      lcd.setCursor(0, 0); lcd.print(g1.substring(0, 16)); 

      pos1++; if(pos1 >= m1.length()) pos1 = 0; 

    } 



 

    // Alt Satır: Modlar, Programlar ve Menü 

    if(millis() - kMillis2 > 350) { 

      kMillis2 = millis(); 

      String durumlar = aktifDurumlariAl(); 

      String m2 = (durumlar != "" ? durumlar + "| " : "") + "A:Prog B:Sil 2512:Saat 2323:Nem 2424:Isi 

0:Acil   "; 

      String g2 = m2.substring(pos2) + m2.substring(0, pos2); 

      lcd.setCursor(0, 1); lcd.print(g2.substring(0, 16)); 

      pos2++; if(pos2 >= m2.length()) pos2 = 0; 

    } 

  } 

 

  char key = keypad.getKey(); 

  if(key) { 

    tusKaydi += key; if(tusKaydi.length() > 6) tusKaydi.remove(0, 1); 

     

    if(tusKaydi == "252525") { lcd.clear(); lcd.print(F("SIFIRLANDI")); for(int i=0; i<512; i++) 

EEPROM.write(i, 0); delay(1500); resetFunc(); } 

    if(tusKaydi == "252512") {  

      int yg = veriGirisi("Gun (1-7):", 1, 1, 7); int ys = veriGirisi("Saat:", 2, 0, 23); int yd = veriGirisi("Dak:", 

2, 0, 59); 

      if(yg!=-1) { gun=yg; saat=ys; dakika=yd; saniye=0; sistemKontrol(); } tusKaydi="";  

    } 

    if(tusKaydi == "242424") {  

      int v = veriGirisi("Isi Esik:", 2, 0, 50); if(v!=-1){sicaklikEsik=v; sicaklikModu=true;} else 

sicaklikModu=false; tusKaydi="";  

    } 

    if(tusKaydi == "232323") {  

      int v = veriGirisi("Nem Esik:", 2, 0, 99); if(v!=-1){nemEsik=v; nemModu=true;} else 

nemModu=false; tusKaydi="";  

    } 

     



    if(key == 'A') { 

      int r = veriGirisi("Role Sec(1-2):", 1, 1, 2); int p = veriGirisi("Prog No(1-5):", 1, 1, 5); 

      int ag = veriGirisi("Ac Gun(1-7):", 1, 1, 7); int as = veriGirisi("Ac Saat:", 2, 0, 23); int ad = 

veriGirisi("Ac Dak:", 2, 0, 59); 

      int kg = veriGirisi("Kap Gun(1-7):", 1, 1, 7); int ks = veriGirisi("Kap Saat:", 2, 0, 23); int kd = 

veriGirisi("Kap Dak:", 2, 0, 59); 

      if(r!=-1) {  

        programlar[r-1][p-1] = {ag, as, ad, kg, ks, kd, true};  

        EEPROM.put(((r-1)*5+(p-1))*sizeof(Alarm), programlar[r-1][p-1]);  

      } 

      sistemKontrol(); 

    } 

    if(key == '0') { digitalWrite(roleler[0], LOW); digitalWrite(roleler[1], LOW); } 

    if(key == '*') { digitalWrite(roleler[0], !digitalRead(roleler[0])); } 

    if(key == '#') { digitalWrite(roleler[1], !digitalRead(roleler[1])); } 

    if(key == 'B') { 

       int r = veriGirisi("Sil Role(1-2):", 1, 1, 2); int p = veriGirisi("Sil Prog(1-5):", 1, 1, 5); 

       if(r != -1 && p != -1) {  

         programlar[r-1][p-1].aktif = false;  

         EEPROM.put(((r-1)*5+(p-1))*sizeof(Alarm), programlar[r-1][p-1]);  

       } 

       sistemKontrol(); 

    } 

  } 

} 

 


